Program code Simulation Model of Dyadic Interaction



4

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'       HERE'S THE CODE FOR THE MAIN SIMULATION PROGRAM (go to line 170 for main loop)
'       THE PROGRAM BRANCHES TO SEVERAL SUB-ROUTINES

'
'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'

Sub new_model()

'

'

'-------THE FIRST PART OF THE MODEL CODE CONTAINS INITIALIZATION ACTIONS

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'-------Clear all values in columns 6 to 17 and set color to None

    Range(Cells(5, 6), Cells(65536, 17)).ClearContents

    Range(Cells(5, 6), Cells(5000, 9)).Interior.ColorIndex = xlNone

'

'

'-------SPECIFY THE VALUE OF THE PARAMETERS

'-------specify number of steps based on value in worksheet

'-------and redefine behavior array

    number_of_steps = Range("number_of_steps").Value

    ReDim array_behavior(number_of_steps, 4) '4 columns refer to two concerns for two children

    ReDim array_saturation(number_of_steps, 2)

    ReDim array_concern_real(number_of_steps, 8)

    ReDim array_randomfactor(number_of_steps, 4) ' this array to check the random factor

'specify the values of the parameters based on the values in the worksheet

    child1_involvement_parameter = Range("child1_involvement").Value

    child1_isolation_parameter = Range(" child1_isolation").Value

    child1_concern_weight = Range("child1_concern_weight").Value

    child2_involvement_parameter = Range("child2_involvement").Value

    child2_isolation_parameter = Range("child2_isolation").Value

    child2_concern_weight = Range("child2_concern_weight").Value

    child1_symmetry_behavior = Range("child1_symmetry_behavior").Value

    child1_continuity_behavior = Range("child1_continuity_behavior").Value

    child2_symmetry_behavior = Range("child2_symmetry_behavior").Value

    child2_continuity_behavior = Range("child2_continuity_behavior").Value

    child1_symmetry_emotion = Range("child1_symmetry_emotion").Value

    child1_continuity_emotion = Range("child1_continuity_emotion").Value

    child2_symmetry_emotion = Range("child2_symmetry_emotion").Value

    child2_continuity_emotion = Range("child2_continuity_emotion").Value

    reward_inv1 = Range("reward_inv1").Value

    reward_iso1 = Range("reward_iso1").Value

    reward_inv2 = Range("reward_inv2").Value

    reward_iso2 = Range("reward_iso2").Value

    ch1_par1_pos = Range("ch1_par1_pos").Value

    ch1_par2_pos = Range("ch1_par2_pos").Value

    ch1_par3_pos = Range("ch1_par3_pos").Value

    ch1_par4_pos = Range("ch1_par4_pos").Value

    ch1_par1_neg = Range("ch1_par1_neg").Value

    ch1_par2_neg = Range("ch1_par2_neg").Value

    ch1_par3_neg = Range("ch1_par3_neg").Value

    ch1_par4_neg = Range("ch1_par4_neg").Value

    ch2_par1_pos = Range("ch2_par1_pos").Value

    ch2_par2_pos = Range("ch2_par2_pos").Value

    ch2_par3_pos = Range("ch2_par3_pos").Value

    ch2_par4_pos = Range("ch2_par4_pos").Value

    ch2_par1_neg = Range("ch2_par1_neg").Value

    ch2_par2_neg = Range("ch2_par2_neg").Value

    ch2_par3_neg = Range("ch2_par3_neg").Value

    ch2_par4_neg = Range("ch2_par4_neg").Value

    ch_weight_neutral = Range("ch_weight_neutral").Value

    ch_weight_cont = Range("ch_weight_cont").Value

    ch_weight_sym = Range("ch_weight_sym").Value

    meanpos_child1 = Range("meanpos_child1").Value

    meanneg_child1 = Range("meanneg_child1").Value

    meanpos_child2 = Range("meanpos_child2").Value

    meanneg_child2 = Range("meanneg_child2").Value

    weight_overrealized_concern = Range("weight_overrealized_concern").Value   '  0.5

    randomization_ratio = Range("randomization_ratio").Value

    memory = Range("memory").Value

    ReDim array_memory_involvement_1(memory) 'redefine arrays with length equal to variable "memory"

    ReDim array_memory_isolation_1(memory)

    ReDim array_memory_involvement_2(memory)

    ReDim array_memory_isolation_2(memory)

    child1_involvement_variable = child1_involvement_parameter

    child1_isolation_variable = child1_isolation_parameter

    child2_involvement_variable = child2_involvement_parameter

    child2_isolation_variable = child2_isolation_parameter

'

'

'-------INITIALIZE THE VALUE OF THE CONCERN WEIGHTS AND REALIZATION VALUES

    child1_involvement_realised = 0

    child2_involvement_realised = 0

    child1_isolation_realised = 0

    child2_isolation_realised = 0

'

'-------INITIALIZE THE VALUE OF THE FIRST BEHAVIOR

'

If child1_involvement_parameter >= child1_isolation_parameter Then

        array_behavior(1, 1) = "together"

    Else

        array_behavior(1, 1) = "alone"

End If

If child2_involvement_parameter >= child2_isolation_parameter Then

        array_behavior(1, 2) = "together"

    Else

        array_behavior(1, 2) = "alone"

End If

'

'-------INITIAL CALCULATION OF CONCERN REALIZATION FOR i=1 AND plaats = 1

'

    i = 1

    plaats = 1

'

'-------BRANCH TO SUBROUTINE 1 CALCULATE CONCERN REALIZATION (go to line 343)
Call calculate_concern_realization

'

'-------USE INFORMATION FROM SUBROUTINE TO SPECIFY VALUES OF VARIABLES

    child1_involvement_realised = Application.Sum(array_memory_involvement_1)

    child2_involvement_realised = Application.Sum(array_memory_involvement_2)

    child1_isolation_realised = Application.Sum(array_memory_isolation_1)

    child2_isolation_realised = Application.Sum(array_memory_isolation_2)

'

'-------CALCULATE RATIOS OF REALIZED CONCERNS

'-------since there is only 1 event so far, the ratios are equal to the level realised

    child1_involvement_ratio_realised = child1_involvement_realised

    child2_involvement_ratio_realised = child2_involvement_realised

    child1_isolation_ratio_realised = child1_isolation_realised

    child2_isolation_ratio_realised = child2_isolation_realised

'

'-------CALCULATE INITIAL DRIVES

'-------calculate the initial drives as the difference between the ratio preferred and the ratio realised

    child1_involvement_drive = child1_involvement_preferred - child1_involvement_ratio_realised

    child2_involvement_drive = child2_involvement_preferred - child2_involvement_ratio_realised

    child1_isolation_drive = child1_isolation_preferred - child1_isolation_ratio_realised

    child2_isolation_drive = child2_isolation_preferred - child2_isolation_ratio_realised

'

'-------BRANCH TO SUBROUTINE 2: CALCULATE EMOTION LEVEL go to line 409
Call calculate_emotion_level

'

'-------SPECIFY EMOTIONAL EXPRESSION

'-------no call to macro: the emotional expression of the first time unit is set to neutral

    array_behavior(1, 3) = "neutral"

    array_behavior(1, 4) = "neutral"

'

'-------INITIALIZE THE RANDOM VALUES FOR randomize_together AND randomize_alone

'-------these variables are used in the macro calculation_random_factors

If randomization_ratio > 0 Then

    rand_randomize_together_mean1 = 0.5

    rand_randomize_alone_mean1 = 0.5

    rand_randomize_together_mean2 = 0.5

    rand_randomize_alone_mean2 = 0.5

End If

'

'-------INITIALIZE EMOTION DURATION PARAMETERS

    emotion_duration_pos_ch1 = 0

    emotion_duration_neg_ch1 = 0

    emotion_duration_pos_ch2 = 0

    emotion_duration_neg_ch2 = 0

'

'

'-------MAIN PROGRAM ROUTINE (LOOP FROM STEP 2 TO TOTAL NUMBER OF STEPS)

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'-------BEGIN OF MAIN PROGRAM LOOP

For i = 2 To number_of_steps

'

'-------SPECIFY INFORMATION IN STATUS BAR ABOUT STEP

    Application.StatusBar = "This is run " & i

'

'-------BRANCH TO THIRD SUBROUTINE: CALCULATE PREFERENCE CONCERNS go to line 440
    Call calculate_preference_concerns

'

'-------set weight parameter to calculated value of drive

'-------for child 1

        child1_together_weight = child1_involvement_drive

        child1_alone_weight = child1_isolation_drive

'-------for child 2

        child2_together_weight = child2_involvement_drive

        child2_alone_weight = child2_isolation_drive

'

'-------BRANCH TO FOURTH SUBROUTINE CALCULATE SYMMETRY AND CONTINUITY go to line 478
    Call calculate_symmetrie_and_continuity

'

'-------BRANCH TO FIFTH SUBROUTINE CALCULATE RANDOM FACTORS go to line 520
    Call calculate_random_factors

'

'-------use information from subroutine to specify variables

'-------for child 1

        child1_together_weight = child1_together_weight + randomize_together1

        child1_alone_weight = child1_alone_weight + randomize_alone1

'-------for child 2

        child2_together_weight = child2_together_weight + randomize_together2

        child2_alone_weight = child2_alone_weight + randomize_alone2

'

'-------DETERMINE THE NEXT BEHAVIORAL OUTPUT ON THE BASIS OF THE WEIGHTS

'-------for child 1

    If child1_together_weight > child1_alone_weight Then

        array_behavior(i, 1) = "together"

    Else

        array_behavior(i, 1) = "alone"

    End If

'-------for child 2

    If child2_together_weight > child2_alone_weight Then

        array_behavior(i, 2) = "together"

    Else

        array_behavior(i, 2) = "alone"

    End If

'

'-------CALCULATE THE LEVEL OF CONCERN REALIZATION

'

'-------BRANCH TO SUBROUTINE 1 go to line 343
    Call calculate_concern_realization

'

'-------CALCULATE THE VALUE OF THE DRIVES

'-------by subtracting the realized from the preferred level

        child1_involvement_drive = (child1_involvement_preferred - child1_involvement_ratio_realised)

        child2_involvement_drive = (child2_involvement_preferred - child2_involvement_ratio_realised)

        child1_isolation_drive = (child1_isolation_preferred - child1_isolation_ratio_realised)

        child2_isolation_drive = (child2_isolation_preferred - child2_isolation_ratio_realised)

'

'-------BRANCH TO SUBROUTINE 2 CALCULATE EMOTION LEVEL go to line 411
    Call calculate_emotion_level

'

'-------BRANCH TO SUBROUTINE 6 CALCULATE EMOTIONAL EXPRESSION go to line 574
    Call calculate_emotional_expression

'

'-------WRITE RESULT OF SUBROUTINE TO ARRAYS

        array_behavior(i, 3) = child1_emotional_expression

        array_behavior(i, 4) = child2_emotional_expression

'

'-------CALCULATE EFFECT OF EXPRESSION ON PREFERENCE OF CONCERNS

'-------BRANCH TO SUBROUTINE 7 go to line 864
    Call calculate_involvement_and_isolation_variables

'

'-------WRITE RESULT OF SUBROUTINE TO ARRAYS

        array_concern_real(i, 3) = child1_emotions

        array_concern_real(i, 4) = child2_emotions

        array_concern_real(i, 5) = child1_involvement_variable / (child1_involvement_variable + child1_isolation_variable) 'child1_emotions

        array_concern_real(i, 6) = child1_isolation_variable / (child1_involvement_variable + child1_isolation_variable) 'child2_emotions

        array_concern_real(i, 7) = child2_involvement_variable / (child2_involvement_variable + child2_isolation_variable)  'child1_emotions

        array_concern_real(i, 8) = child2_isolation_variable / (child2_involvement_variable + child2_isolation_variable)

'

'-------END OF MAIN PROGRAM LOOP

Next i

'

'

'-------WRITE RESULTS OF MAIN PROGRAM ROUTINE TO WORKSHEET

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'

'-------COPY VALUES OF ARRAY_BEHAVIOR TO WORKSHEET

'-------specify first cell of range_behavior

    Set range_behavior = Range("F5")

'-------specify range for concerns relative to range behavior cell

    Set range_concern_real = Range(range_behavior.Offset(0, 4), range_behavior.Offset(number_of_steps - 1, 11))

'-------specify complete range for range behavior

    Set range_behavior = Range(range_behavior, range_behavior.Offset(number_of_steps - 1, 3))

'-------specify ranges for emotion and expression

    Set range_emotion = Range(range_behavior.Offset(0, 2), range_behavior.Offset(number_of_steps - 1, 3))

    Set range_emotion_intensity = Range(range_behavior.Offset(0, 4), range_behavior.Offset(number_of_steps - 1, 5))

'-------clear contents and color codes of range

    With range_behavior

        .Font.ColorIndex = 0

        ' 0 is wit

        .Interior.ColorIndex = xlNone

    End With

    With range_emotion_intensity

        .Font.ColorIndex = 0

        ' 0 is wit

        .Interior.ColorIndex = xlNone

    End With

'

'-------copy values to worksheet

    range_behavior.Value = array_behavior

    range_concern_real.Value = array_concern_real

'

'-------change interior and font colors in range_behavior dependent on value of cells

    For Each my_cell In range_behavior

        If my_cell.Value = "alone" Then

            my_cell.Interior.ColorIndex = 10 ' 43

            my_cell.Font.ColorIndex = 2

        End If

        If my_cell.Value = "together" Then my_cell.Interior.ColorIndex = 6

        If my_cell.Value = "positive" Then

            my_cell.Interior.ColorIndex = 3

            my_cell.Font.ColorIndex = 2

        End If

        If my_cell.Value = "neutral" Then my_cell.Interior.ColorIndex = 42

        If my_cell.Value = "negative" Then

            my_cell.Interior.ColorIndex = 11  ' 39

            my_cell.Font.ColorIndex = 2

        End If

    Next my_cell

'

'-------copy emotion range

    range_emotion.Copy

    range_emotion_intensity.PasteSpecial Paste:=xlFormats, Operation:=xlNone, SkipBlanks:= _

        False, Transpose:=False

    Application.CutCopyMode = False

'-------specify variable for eventual repeated run

        number_of_steps_old = number_of_steps

'-------check effect of random variable

        Range(Cells(5, 21), Cells(number_of_steps + 4, 24)).Value = array_randomfactor

'-------reset statusbar information

    Application.StatusBar = False

    Columns("L:N").Font.ColorIndex = 2

    Range("t7").Select

'

'-------END OF THE CODE FOR THE MAIN SIMULATION PROGRAM.

End Sub

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'       SUBROUTINES FOR THE MAIN SIMULATION PROGRAM

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 1

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_concern_realization()

'-------DETERMINE VARIABLES DEPENDENT ON LENGTH OF MEMORY

    If i <= memory Then

        deler = i

        plaats = i

    Else

        deler = memory

        plaats = memory

    End If

'

'-------SHIFT CONTENTS OF MEMORY ONE POSITION BACK

    If i > memory Then

        For j = 1 To memory - 1

            array_memory_involvement_1(j) = array_memory_involvement_1(j + 1)

            array_memory_isolation_1(j) = array_memory_isolation_1(j + 1)

            array_memory_involvement_2(j) = array_memory_involvement_2(j + 1)

            array_memory_isolation_2(j) = array_memory_isolation_2(j + 1)

        Next j

    End If

'

'-------DETERMINE CONCERN REALIZATION

    If array_behavior(i, 1) = "together" And array_behavior(i, 2) = "together" Then

        array_memory_involvement_1(plaats) = reward_inv1

        array_memory_involvement_2(plaats) = reward_inv2

        array_memory_isolation_1(plaats) = 0

        array_memory_isolation_2(plaats) = 0

    End If

    If array_behavior(i, 1) = "together" And array_behavior(i, 2) = "alone" Then

        array_memory_involvement_1(plaats) = 0

        array_memory_involvement_2(plaats) = 0

        array_memory_isolation_1(plaats) = 0

        array_memory_isolation_2(plaats) = (reward_iso2) / 2

    End If

    If array_behavior(i, 2) = "together" And array_behavior(i, 1) = "alone" Then

        array_memory_involvement_1(plaats) = 0

        array_memory_involvement_2(plaats) = 0

        array_memory_isolation_1(plaats) = (reward_iso1) / 2

        array_memory_isolation_2(plaats) = 0

    End If

    If array_behavior(i, 1) = "alone" And array_behavior(i, 2) = "alone" Then

        array_memory_involvement_1(plaats) = 0

        array_memory_involvement_2(plaats) = 0

        array_memory_isolation_1(plaats) = reward_iso1

        array_memory_isolation_2(plaats) = reward_iso2

    End If

'-------DETERMINE THE RATIO OF CONCERN REALIZATION

'-------make sum of "rewards" from memory

    child1_involvement_realised = Application.Sum(array_memory_involvement_1)

    child2_involvement_realised = Application.Sum(array_memory_involvement_2)

    child1_isolation_realised = Application.Sum(array_memory_isolation_1)

    child2_isolation_realised = Application.Sum(array_memory_isolation_2)

'------divide sum of rewards by number of time steps made so far

    child1_involvement_ratio_realised = child1_involvement_realised / deler

    child2_involvement_ratio_realised = child2_involvement_realised / deler

    child1_isolation_ratio_realised = child1_isolation_realised / deler

    child2_isolation_ratio_realised = child2_isolation_realised / deler

'

'-------END OF SUBROUTINE 1 go back to line 123 or to line 220
End Sub

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 2

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_emotion_level()

'-------DEFINE VARIABLES NEEDED WITHIN THIS SOUBROUTINE

    Dim gewicht_inv1, gewicht_iso1

    Dim gewicht_inv2, gewicht_iso2

    Dim W

    weight_overrealized_concern = Range("weight_overrealized_concern").Value   '  0.5

    W = weight_overrealized_concern

'

'-------CALCULATE EMOTION LEVELS

    If child1_involvement_drive < 0 Then gewicht_inv1 = W Else gewicht_inv1 = 1

    If child1_isolation_drive < 0 Then gewicht_iso1 = W Else gewicht_iso1 = 1

    If child2_involvement_drive < 0 Then gewicht_inv2 = W Else gewicht_inv2 = 1

    If child2_isolation_drive < 0 Then gewicht_iso2 = W Else gewicht_iso2 = 1

    child1_emotions = (Abs(child1_involvement_drive) * gewicht_inv1 + Abs(child1_isolation_drive) * gewicht_iso1) / (gewicht_inv1 + gewicht_iso1)

    child2_emotions = (Abs(child2_involvement_drive) * gewicht_inv2 + Abs(child2_isolation_drive) * gewicht_iso2) / (gewicht_inv2 + gewicht_iso2)

    child1_emotions = (child1_emotions * -2) + 1

    child2_emotions = (child2_emotions * -2) + 1

'-------END OF SUBROUTINE 2 go back to line 146 or to line 230
End Sub

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 3

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_preference_concerns()

'-------calculates the prefered state as a proportion on the basis of the value of the

'-------concerns, based on the positive or negative emotional expression of the preceding time step

'-------which was calculated in the macro calculate involvement_and_isolation_variables).

'

    If child1_involvement_variable < 0 Then child1_involvement_variable = 0

    If child1_isolation_variable < 0 Then child1_isolation_variable = 0

    If child2_involvement_variable < 0 Then child2_involvement_variable = 0

    If child2_isolation_variable < 0 Then child2_isolation_variable = 0

    If child1_involvement_variable + child1_isolation_variable = 0 Then

        child1_involvement_preferred = 0

        child1_isolation_preferred = 0

    Else

        child1_involvement_preferred = child1_involvement_variable / (child1_involvement_variable + child1_isolation_variable)

        child1_isolation_preferred = child1_isolation_variable / (child1_involvement_variable + child1_isolation_variable)

    End If

    If child2_involvement_variable + child2_isolation_variable = 0 Then

        child2_involvement_preferred = 0

        child2_isolation_preferred = 0

    Else

        child2_involvement_preferred = child2_involvement_variable / (child2_involvement_variable + child2_isolation_variable)

        child2_isolation_preferred = child2_isolation_variable / (child2_involvement_variable + child2_isolation_variable)

    End If

'       END OF SUBROUTINE 3 go back to line 179
End Sub

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 4

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_symmetrie_and_continuity()

'-------DETERMINE VALUES OF VARIABLES DEPENDENT ON PRECEDING BEHAVIOR

    If array_behavior(i - 1, 1) = "together" Then

        child1_together_cont = child1_continuity_behavior

        child2_together_sym = child2_symmetry_behavior

        child1_alone_cont = 0

        child2_alone_sym = 0

    End If

    If array_behavior(i - 1, 1) = "alone" Then

       child1_alone_cont = child1_continuity_behavior

       child2_alone_sym = child2_symmetry_behavior

       child1_together_cont = 0

       child2_together_sym = 0

    End If

    If array_behavior(i - 1, 2) = "together" Then

       child1_together_sym = child1_symmetry_behavior

       child2_together_cont = child2_continuity_behavior

       child1_alone_sym = 0

       child2_alone_cont = 0

    End If

    If array_behavior(i - 1, 2) = "alone" Then

       child1_alone_sym = child1_symmetry_behavior

       child2_alone_cont = child2_continuity_behavior

       child1_together_sym = 0

       child2_together_cont = 0

    End If

'

'-------CALCULATE WEIGHTS FOR BEHAVIORS

    child1_together_weight = child1_together_weight + child1_together_cont + child1_together_sym

    child2_together_weight = child2_together_weight + child2_together_cont + child2_together_sym

    child1_alone_weight = child1_alone_weight + child1_alone_cont + child1_alone_sym

    child2_alone_weight = child2_alone_weight + child2_alone_cont + child2_alone_sym

'

'-------END OF SUBROUTINE 4 go back to line 190
End Sub

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 5

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_random_factors()

'-------random factors are based on an iterative random function (a logistic growth

'-------model with random growth factor plus additional random component); the iterative

'-------function guarantees that the random factors are autocorrelated in time

'-------DEFINE VARIABLES NEEDED WITHIN THIS SOUBROUTINE

    Dim rate1, rate2

    Dim rr

    Dim out

    out = 1

    rr = randomization_ratio * 10

'-------CALCULATE RANDOMIZATION FOR CHILD 1

    If randomization_ratio = 0 Then

                randomize_together1 = 0

                randomize_alone1 = 0

    Else

                rate1 = Application.NormInv(Rnd, 0, randomization_ratio)

                rate2 = Application.NormInv(Rnd, 0, randomization_ratio)

                rand_randomize_together_mean1 = rand_randomize_together_mean1 * (1 + rate1 * (1 - rand_randomize_together_mean1))

                randomize_together1 = rand_randomize_together_mean1 + Application.NormInv(Rnd, 0, randomization_ratio) * out - 0.5

                rand_randomize_alone_mean1 = rand_randomize_alone_mean1 * (1 + rate2 * (1 - rand_randomize_alone_mean1))

                randomize_alone1 = rand_randomize_alone_mean1 + Application.NormInv(Rnd, 0, randomization_ratio) * out - 0.5

    End If

'

'-------CALCULATE RANDOMIZATION FOR CHILD 2

    If randomization_ratio = 0 Then

                randomize_together2 = 0

                randomize_alone2 = 0

    Else

                rate1 = Application.NormInv(Rnd, 0, randomization_ratio)

                rate2 = Application.NormInv(Rnd, 0, randomization_ratio)

                rand_randomize_together_mean2 = rand_randomize_together_mean2 * (1 + rate1 * (1 - rand_randomize_together_mean2))

                randomize_together2 = rand_randomize_together_mean2 + Application.NormInv(Rnd, 0, randomization_ratio) * out - 0.5

                rand_randomize_alone_mean2 = rand_randomize_alone_mean2 * (1 + rate2 * (1 - rand_randomize_alone_mean2))

                randomize_alone2 = rand_randomize_alone_mean2 + Application.NormInv(Rnd, 0, randomization_ratio) * out - 0.5

    End If

'-------WRITE THE RANDOM FACTORS TO AN ARRAY

        array_randomfactor(i, 1) = randomize_together1

        array_randomfactor(i, 2) = randomize_alone1

        array_randomfactor(i, 3) = randomize_together2

        array_randomfactor(i, 4) = randomize_alone2

'

'-------END OF SUBROUTINE 5 go back to line 193
End Sub

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 6

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_emotional_expression()

'

'-------DEFINE VARIABLES NEEDED WITHIN THIS SOUBROUTINE

    Dim array_expression_weights(3, 2)

'-------In this array, the first row is for positive, the second for neutral and the

'-------third for negative expressions; column 1 is for child 1, column 2 is for child 2

    Dim c_1 'counters for emorional expression loops

    Dim c_2

    Dim max_weight_pos_neg ' maximum weight factor

    Dim randomization_ratio_emotion 'regulates the strength of the random influence

    Dim t 'temporary variable value

'

'-------define values for variables from worksheet

    randomization_ratio_emotion = Range("randomization_ratio_emotion").Value

'

'-------CALCULATE WEIGHTS OR PROBABILITIES FOR EMOTIONAL EXPRESSION

'-------for child 1

    array_expression_weights(1, 1) = my_sigmoid(child1_emotions, _

                                ch1_par1_pos, _

                                ch1_par2_pos, _

                                ch1_par3_pos, _

                                ch1_par4_pos)

    array_expression_weights(3, 1) = my_sigmoid(child1_emotions, _

                                ch1_par1_neg, _

                                ch1_par2_neg, _

                                ch1_par3_neg, _

                                ch1_par4_neg)

    If array_expression_weights(1, 1) >= array_expression_weights(3, 1) Then

            max_weight_pos_neg = array_expression_weights(1, 1)

        Else

            max_weight_pos_neg = array_expression_weights(3, 1)

    End If

    array_expression_weights(2, 1) = ch_weight_neutral - max_weight_pos_neg

'

'-------for child 2

    array_expression_weights(1, 2) = my_sigmoid(child2_emotions, _

                                    ch2_par1_pos, _

                                    ch2_par2_pos, _

                                    ch2_par3_pos, _

                                    ch2_par4_pos)

    array_expression_weights(3, 2) = my_sigmoid(child2_emotions, _

                                    ch2_par1_neg, _

                                    ch2_par2_neg, _

                                    ch2_par3_neg, _

                                    ch2_par4_neg)

    If array_expression_weights(1, 2) >= array_expression_weights(3, 2) Then

            max_weight_pos_neg = array_expression_weights(1, 2)

        Else

            max_weight_pos_neg = array_expression_weights(3, 2)

    End If

    array_expression_weights(2, 2) = ch_weight_neutral - max_weight_pos_neg

'

'-------draw a random value on the basis of these weights and put value in array

    For c_1 = 1 To 3

        For c_2 = 1 To 2

            t = array_expression_weights(c_1, c_2)

            array_expression_weights(c_1, c_2) = t - t * randomization_ratio_emotion + t * randomization_ratio_emotion * Rnd

        Next c_2

    Next c_1

'

'-------add weights based on continuity and symmetry variables

    If array_behavior(i - 1, 3) = "positive" Then

        array_expression_weights(1, 1) = array_expression_weights(1, 1) + child1_continuity_emotion

        array_expression_weights(1, 2) = array_expression_weights(1, 2) + child2_symmetry_emotion

    End If

    If array_behavior(i - 1, 4) = "positive" Then

        array_expression_weights(1, 2) = array_expression_weights(1, 2) + child2_continuity_emotion

        array_expression_weights(1, 1) = array_expression_weights(1, 1) + child1_symmetry_emotion

    End If

    If array_behavior(i - 1, 3) = "neutral" Then

        array_expression_weights(2, 1) = array_expression_weights(2, 1) + child1_continuity_emotion

        array_expression_weights(2, 2) = array_expression_weights(2, 2) + child2_symmetry_emotion

    End If

    If array_behavior(i - 1, 4) = "neutral" Then

        array_expression_weights(2, 2) = array_expression_weights(2, 2) + child2_continuity_emotion

        array_expression_weights(2, 1) = array_expression_weights(2, 1) + child1_symmetry_emotion

    End If

'

    If array_behavior(i - 1, 3) = "negative" Then

        array_expression_weights(3, 1) = array_expression_weights(3, 1) + child1_continuity_emotion

        array_expression_weights(3, 2) = array_expression_weights(3, 2) + child2_symmetry_emotion

    End If

    If array_behavior(i - 1, 4) = "negative" Then

        array_expression_weights(3, 2) = array_expression_weights(3, 2) + child2_continuity_emotion

        array_expression_weights(3, 1) = array_expression_weights(3, 1) + child1_symmetry_emotion

    End If

'

'-------calculate proportions on the basis of these weights

'-------for child 1

    emotie_proportie_pos_ch1 = array_expression_weights(1, 1) / (array_expression_weights(2, 1) + array_expression_weights(3, 1))

    emotie_proportie_neut_ch1 = (array_expression_weights(1, 1) + array_expression_weights(3, 1)) / array_expression_weights(2, 1)

'---note that for neutral the proportion must be calculated inversely, namely pos plus neg, divided by neut

    emotie_proportie_neg_ch1 = array_expression_weights(3, 1) / (array_expression_weights(2, 1) + array_expression_weights(1, 1))

'

'-------for child 2

    emotie_proportie_pos_ch2 = array_expression_weights(1, 2) / (array_expression_weights(2, 2) + array_expression_weights(3, 2))

    emotie_proportie_neut_ch2 = (array_expression_weights(1, 2) + array_expression_weights(3, 2)) / array_expression_weights(2, 2)

    emotie_proportie_neg_ch2 = array_expression_weights(3, 2) / (array_expression_weights(2, 2) + array_expression_weights(1, 2))

'

'-------FOR CHILD 1

'-------determine which emotion has the greatest weight

        max_weight_pos_neg = array_expression_weights(1, 1)

'

    If array_expression_weights(2, 1) > max_weight_pos_neg Then _

                        max_weight_pos_neg = array_expression_weights(2, 1)

    If array_expression_weights(3, 1) > max_weight_pos_neg Then _

                        max_weight_pos_neg = array_expression_weights(3, 1)

'

'-------determine which of the cells carries the greatest weight to determine the

'-------value of max_weight_pos_neg

    If array_expression_weights(1, 1) = max_weight_pos_neg Then t = 1

    If array_expression_weights(2, 1) = max_weight_pos_neg Then t = 2

    If array_expression_weights(3, 1) = max_weight_pos_neg Then t = 3

'

        max_weight_pos_neg = t

'

'-------determine duration of emotional expression for more intensive emotions

'-------which increases influence of stronger emotions

'-------if duration > 0 then carry out behavior governed by duration

'-------if duration = 0 then carry out behavior with greatest weight and

'-------determine new duration for behavior with greatest weight

    If emotion_duration_pos_ch1 > 0 Then

        child1_emotional_expression = "positive"

        emotion_duration_pos_ch1 = emotion_duration_pos_ch1 - 1

        array_concern_real(i, 1) = array_concern_real(i - 1, 1)

    End If

'

    If emotion_duration_neg_ch1 > 0 Then

        child1_emotional_expression = "negative"

        emotion_duration_neg_ch1 = emotion_duration_neg_ch1 - 1

        array_concern_real(i, 1) = array_concern_real(i - 1, 1)

    End If

'

    If emotion_duration_pos_ch1 = 0 And emotion_duration_neg_ch1 = 0 Then

        If max_weight_pos_neg = 1 Then child1_emotional_expression = "positive"

        If max_weight_pos_neg = 2 Then child1_emotional_expression = "neutral"

        If max_weight_pos_neg = 3 Then child1_emotional_expression = "negative"

        If max_weight_pos_neg = 1 Then

            emotion_duration_pos_ch1 = Int(emotie_proportie_pos_ch1)

            '-------Determine intensity category of emotion

            Select Case emotie_proportie_pos_ch1

                Case Is < 2.875

                    array_concern_real(i, 1) = 2 'numbers mean expression codes

                Case Is < 5.25

                    array_concern_real(i, 1) = 3

                Case Is < 7.625

                    array_concern_real(i, 1) = 4

                Case Is >= 7.625

                    array_concern_real(i, 1) = 5

                Case Else

                    array_concern_real(i, 1) = "Fout!"    'means error

            End Select

        End If

        If max_weight_pos_neg = 2 Then

            '-------determine intensity category of emotion

            Select Case emotie_proportie_neut_ch1

                Case Is < 0.44

                    array_concern_real(i, 1) = 0    'intensity code for neutral is 0

                Case Else

                    If array_expression_weights(3, 1) < array_expression_weights(1, 1) Then

                        array_concern_real(i, 1) = 1    'if the negative value is smaller than the positive value

                                                        'is he neutral expression slightly positive, hence 1

                    End If

                    If array_expression_weights(1, 1) < array_expression_weights(3, 1) Then

                        array_concern_real(i, 1) = -1       'if the positive value is smaller than the negative value

                                                            'the neutral expression is slightly negative, hence -1

                    End If

            End Select

        End If

        If max_weight_pos_neg = 3 Then

            emotion_duration_neg_ch1 = Int(emotie_proportie_neg_ch1)

            '-------determine intensity category of emotion

            Select Case emotie_proportie_neg_ch1

                Case Is < 2.8

                    array_concern_real(i, 1) = -2    'emotion expression code

                Case Is < 5.1

                    array_concern_real(i, 1) = -3

                Case Is >= 5.1

                    array_concern_real(i, 1) = -4

                Case Else

                    array_concern_real(i, 1) = "Fout!"    'means error

            End Select

        End If

    End If

'

'-------FOR CHILD 2

'-------determine which emotion has the greatest weight

        max_weight_pos_neg = array_expression_weights(1, 2)

    If array_expression_weights(2, 2) > max_weight_pos_neg Then _

                        max_weight_pos_neg = array_expression_weights(2, 2)

    If array_expression_weights(3, 2) > max_weight_pos_neg Then _

                        max_weight_pos_neg = array_expression_weights(3, 2)

'

'-------see explanation in child 1

    If array_expression_weights(1, 2) = max_weight_pos_neg Then t = 1

    If array_expression_weights(2, 2) = max_weight_pos_neg Then t = 2

    If array_expression_weights(3, 2) = max_weight_pos_neg Then t = 3

'

        max_weight_pos_neg = t

'

    If emotion_duration_pos_ch2 > 0 Then

        child2_emotional_expression = "positive"

        emotion_duration_pos_ch2 = emotion_duration_pos_ch2 - 1

        array_concern_real(i, 2) = array_concern_real(i - 1, 2)

    End If

'

    If emotion_duration_neg_ch2 > 0 Then

        child2_emotional_expression = "negative"

        emotion_duration_neg_ch2 = emotion_duration_neg_ch2 - 1

        array_concern_real(i, 2) = array_concern_real(i - 1, 2)

    End If

    If emotion_duration_pos_ch2 = 0 And emotion_duration_neg_ch2 = 0 Then

        If max_weight_pos_neg = 1 Then child2_emotional_expression = "positive"

        If max_weight_pos_neg = 2 Then child2_emotional_expression = "neutral"

        If max_weight_pos_neg = 3 Then child2_emotional_expression = "negative"

        If max_weight_pos_neg = 1 Then

            emotion_duration_pos_ch2 = Int(emotie_proportie_pos_ch2)

            Select Case emotie_proportie_pos_ch2

                Case Is < 2.875

                    array_concern_real(i, 2) = 2

                Case Is < 5.25

                    array_concern_real(i, 2) = 3

                Case Is < 7.625

                    array_concern_real(i, 2) = 4

                Case Is >= 7.625

                    array_concern_real(i, 2) = 5

                Case Else

                    array_concern_real(i, 2) = "Fout!"

            End Select

        End If

        If max_weight_pos_neg = 2 Then

            Select Case emotie_proportie_neut_ch2

                Case Is < 0.44

                    array_concern_real(i, 2) = 0

                Case Else

                    If array_expression_weights(3, 2) < array_expression_weights(1, 2) Then

                        array_concern_real(i, 2) = 1

                    End If

                    If array_expression_weights(1, 2) < array_expression_weights(3, 2) Then

                        array_concern_real(i, 2) = -1

                    End If

            End Select

        End If

        If max_weight_pos_neg = 3 Then

            emotion_duration_neg_ch2 = Int(emotie_proportie_neg_ch2)

            Select Case emotie_proportie_neg_ch2

                Case Is < 2.875

                    array_concern_real(i, 2) = -2

                Case Is < 5.25

                    array_concern_real(i, 2) = -3    'emotie expressie code -3

                Case Is >= 5.25

                    array_concern_real(i, 2) = -4

                Case Else

                    array_concern_real(i, 2) = "Fout!"

            End Select

        End If

    End If

'-------END OF SUBROUTINE 6 go back to line 233
End Sub

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       FUNCTION FOR CALCULATING SYGMOID USED IN SUBROUTINE 6

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Function my_sigmoid#(ByVal X#, a, b, c, d)

Application.Volatile

Dim Y#

Y# = a + b / (1# + Exp(-(X# - c) / d))

  my_sigmoid# = Y#

End Function

'

'

'

'

'

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

'       SUBROUTINE 7

'XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

Sub calculate_involvement_and_isolation_variables()

'-------calculates changes in involvement and isolation (old term for the current

'-------autonomy variable) by means of a logistic growth model

'-------DEFINE VARIABLES NEEDED WITHIN THIS SOUBROUTINE

    Dim RatePos_child1

    Dim RateNeg_child1

    Dim RatePos_child2

    Dim RateNeg_child2

    If randomization_ratio = 0 Then

        RatePos_child1 = meanpos_child1

        RateNeg_child1 = meanneg_child1

        RatePos_child2 = meanpos_child2

        RateNeg_child2 = meanneg_child2

    Else

        RatePos_child1 = meanpos_child1

        RateNeg_child1 = meanneg_child1

        RatePos_child2 = meanpos_child2

        RateNeg_child2 = meanneg_child2

    End If

    child1_behavior = array_behavior(i, 1)

    child2_behavior = array_behavior(i, 2)

'

'-------SPECIFY CHANGE IN VARIABLES IN FUNCTION OF PRECEDING BEHAVIORS AND EMOTIONS

'

    If child1_behavior = "together" And child1_emotional_expression = "positive" Then

        child1_involvement_variable = child1_involvement_variable * (1 + RatePos_child1 * (1 - child1_involvement_variable / Range("limit").Value))

        child2_involvement_variable = child2_involvement_variable * (1 + RatePos_child2 * (1 - child2_involvement_variable / Range("limit").Value))

    End If

    If child2_behavior = "together" And child2_emotional_expression = "positive" Then

        child1_involvement_variable = child1_involvement_variable * (1 + RatePos_child1 * (1 - child1_involvement_variable / Range("limit").Value))

        child2_involvement_variable = child2_involvement_variable * (1 + RatePos_child2 * (1 - child2_involvement_variable / Range("limit").Value))

    End If

    If child1_behavior = "together" And child1_emotional_expression = "negative" Then

        child1_involvement_variable = child1_involvement_variable * (1 + RateNeg_child1 * (1 - child1_involvement_variable / Range("limit").Value))

        child2_involvement_variable = child2_involvement_variable * (1 + RateNeg_child2 * (1 - child2_involvement_variable / Range("limit").Value))

    End If

    If child2_behavior = "together" And child2_emotional_expression = "negative" Then

        child1_involvement_variable = child1_involvement_variable * (1 + RateNeg_child1 * (1 - child1_involvement_variable / Range("limit").Value))

        child2_involvement_variable = child2_involvement_variable * (1 + RateNeg_child2 * (1 - child2_involvement_variable / Range("limit").Value))

    End If

    If child1_behavior = "alone" And child1_emotional_expression = "positive" Then

        child1_isolation_variable = child1_isolation_variable * (1 + RatePos_child1 * (1 - child1_isolation_variable / Range("limit").Value))

        child2_isolation_variable = child2_isolation_variable * (1 + RatePos_child2 * (1 - child2_isolation_variable / Range("limit").Value))

    End If

    If child2_behavior = "alone" And child2_emotional_expression = "positive" Then

        child1_isolation_variable = child1_isolation_variable * (1 + RatePos_child1 * (1 - child1_isolation_variable / Range("limit").Value))

        child2_isolation_variable = child2_isolation_variable * (1 + RatePos_child2 * (1 - child2_isolation_variable / Range("limit").Value))

    End If

    If child1_behavior = "alone" And child1_emotional_expression = "negative" Then

        child1_isolation_variable = child1_isolation_variable * (1 + RateNeg_child1 * (1 - child1_isolation_variable / Range("limit").Value))

        child2_isolation_variable = child2_isolation_variable * (1 + RateNeg_child2 * (1 - child2_isolation_variable / Range("limit").Value))

    End If

    If child2_behavior = "alone" And child2_emotional_expression = "negative" Then

        child1_isolation_variable = child1_isolation_variable * (1 + RateNeg_child1 * (1 - child1_isolation_variable / Range("limit").Value))

        child2_isolation_variable = child2_isolation_variable * (1 + RateNeg_child2 * (1 - child2_isolation_variable / Range("limit").Value))

    End If

'-------END OF SUBROUTINE 7 go back to line 241
End Sub

